UR1986_62 - FINAL REPORT -

1986/62. A FORTRAN program for plotting areally distributed gravity (or other) data (Revision 2).

R.G. Richardson

Abstract

This program plots sample points, the data value, and the reading number for data on a metric grid in the range 0 to 1000 km east and north. The plotting scale is specified when the program is run.

THE PROGRAM

STNPLT (Appendix 1)

This program was written for plotting gravity data, but any data in a suitable format can be plotted. The program reads the search parameters and plotting options from logical unit 5 and the data from logical unit 4. A listing of the data values actually plotted is output on logical unit 6 and the statistics of the search are output on logical unit 7. Details of the plotting routines may be found in Richardson (1983).

Control data read from logical unit 5 is:

PWIDTH - V if the AO plotter paper is rotated to have an X-axis 31.4 inches long and a Y-axis 44.3 inches long.
- leave blank for the AO plotter paper to have an X-axis 44.3 inches long and a Y-axis 31.4 inches long.

XMIN - the western boundary of the data (km)
XMAX - the eastern boundary of the data (km)
YMIN - the southern boundary of the data (km)
YMAX - the northern boundary of the data (km)
SCALE - the plotting scale, e.g. 50 000

The labelling used depends on the value of SCALE.

(a) SCALE › 100 000
label each multiple of 10 km
draw a line parallel to the prime axes at each multiple of 100 km

(b) 25 000 ‹ SCALE ‹ 100 000
label each 1 km
draw a line parallel to the prime axes at each multiple of 10 km

(c) 5000 ‹ SCALE ‹ 25 000
label each 250 m
draw a line parallel to the prime axes each 1 km

(d) 2000 ‹ SCALE ‹ 5000
label each 100 m
draw a line parallel to the prime axes each multiple of 500 m

(e) SCALE ‹ 2000
draw a line parallel to the prime axes each multiple of 100 m

If SCALE is such that the map will not plot on the paper width, the ranges and scale may be altered.

IPLCE - the number of decimal places to use for data values written on the plots.

AOK - Y if reduced size symbols are required for close data points, N otherwise.

TITLE - the title for the plot followed by ..

IOPT - select the functions desired from the program.

(a) IOPT = 1
plot symbols only at the data points

(b) IOPT = 2
plot a symbol at each data point and a circle representing the maximum acceptable deviation from the nominal data point. The X and Y separations (XSP, YSP) of the nominal points and the acceptable distance from the nominal point are read. Starting at (XMIN, YMIN) circles are drawn at intervals XSP, YSP until (XMAX, YMAX) is reached. This is a very slow process.

(c) IOPT = 3
plot symbols at the data points and write the data point numbers.

(d) IOPT = 4
plot symbols at the data points and write the data values.

(e) IOPT = 5
plot symbols at the data points and write both the data point numbers and the data values.

AOK - Y if more plots to follow, N otherwise.

Data input from logical unit 4 is:

SURVNO, STATNO, X, Y, BA - format (I5, 1X, I4, 2F9.1, 36X, F8.2)

SURVNO (optional) - up to 14 different survey numbers may be used. A different data point symbol is used for each survey number.

STATNO - the data point number.

X, Y - the east and north co-ordinates of the data point in metres (X, Y in the range 0.0 to 700 000).

BA - the data value.

A summary of the data plotted is output on logical unit 6. The total number of data points in the file on logical unit 4 and the number of points plotted are printed on logical unit 7.

Appendices 2 to 8 show the plotted output of the same set of data for the following control inputs:

Appendix	*XMIN*	*XMAX*	*YMIN*	*YMAX*	*SCALE*	*IPLCE*	*SMALL SYMBOL*	*IOPT*
2	500.5	520.5	294.5	322.5	250 000	0	Y	1
3	500.5	514.5	294.5	312.5	100 000	0	N	3
4	500.5	506.5	294.5	305.5	50 000	1	N	4
5	500.5	503.5	294.5	298.5	25 000	1	N	5
6	500.9	502.2	294.9	296.1	5 000	1	N	5
7	500.9	501.3	294.8	295.3	2 500	1	N	5
8	500.9	501.1	294.8	295.1	2 000	1	N	5

REFERENCE

RICHARDSON, R.G. 1986. Hard copy plotting on the Geological Survey mini-computer (Revision 2). *Unpubl. Rep. Dep. Mines Tasm.* 1986/17.

[9 October 1986]

APPENDIX 1

Program STNPLT

```
$TITL GRAVITY STATION PLOT
C STNPLT,STNPLJ
C PLOTS GRAVITY STATION NUMBERS AND B.A. AT THE STATION LOCATION
C USES A DIFFERENT SYMBOL FOR UP TO 14 SURVEYS.
C ASSUMES COORDINATES ARE IN KM.
C GRAVITY DATA ON LU 4
C CONTROL LU 5
C LIST OF STATIONS ON LU 6
C NUMBER OF STATIONS LU 7
      REAL SCALE,FACT,BA,X,Y,PWIDTH,
     . YMXLST,ERRRAD,XBOT,YBOT,YMXPLT,YTMP
      DOUBLE PRECISION XMIN,XMAX,YMIN,YMAX
      INTEGER*4 IXMIN,IXMAX,IYMIN,IYMAX,SPCE,DFACT,TXSP,DX,DY,
     . IFXMIN,IFXMAX,IFYMIN,IFYMAX,XSP,YSP
      INTEGER*2 NSURV,SURVEY(50),TITLE(40),
     . IOPT,I,SURVNO,STATNO,NTOT,NPLOT,YES,NO,AOK,AR,AS
      LOGICAL SURV,IER,LINE,LABEL,METRES
      DATA YES/1HY/,NO/1HN/,AR/1HR/,AS/1HS/
      OPEN(UNIT=5,FILE='CON:')
C SET TO TAKE INPUT FROM CONSOLE
      YMXPLT=0.
C WIDTH OF PLOTTER PAPER USED
      WRITE(5,864)
  864 FORMAT(' DEFAULT PAPER SIZE HAS XMAX=44.3", YMAX=31.4"'/
     .     ' TYPE V HERE FOR PAPER SIZE XMAX=31.4", YMAX=44.3"')
      READ(5,229) AOK
      PWIDTH=31.4
      IF (AOK .EQ. 'V') PWIDTH=44.3
C Y AXIS IS 31.4 INCHES IF PAPER IS IN DEFAULT DIRECTION
C OR 44.3 INCHES IF HP AXES ARE ROTATED
   21 NTOT=0
      NPLOT=0
C USED TO COUNT THE TOTAL NUMBER OF STATIONS IN THE FILE AND THE
C NUMBER OF STATIONS PLOTTED
  860 WRITE(5,100)
  100 FORMAT(' XMINNNNNN XMAXXXXXX IN KM')
      READ(5,*) XMIN,XMAX
      IF (XMAX .LE. XMIN) GOTO 860
C LOWER AND UPPER X VALUES (KM E)
      XMIN=XMIN*1000.0D00
      IXMIN=XMIN
      XMAX=XMAX*1000.0D00
      IXMAX=XMAX
C CONVERT TO METRES
  861 WRITE(5,102)
  102 FORMAT(' YMINNNNNN YMAXXXXXX IN KM')
      READ(5,*) YMIN,YMAX
      IF (YMAX .LE. YMIN) GOTO 861
C LOWER AND UPPER Y VALUES (KM N)
      YMIN=YMIN*1000.0D00
      IYMIN=YMIN
```

```
      YMAX=YMAX*1000.0D00
      IYMAX=YMAX
C CONVERT TO METRES
  862 WRITE(5,103)
  103 FORMAT(' SCALE?(E.G. 50000.)')
      READ(5,*) SCALE
      FACT=1.E+2/(2.54*SCALE)
C THE FACTOR TO CONVERT FROM GRID METRES TO PLOTTER INCHES
      LABEL=.TRUE.
      METRES=.TRUE.
      IF (SCALE .LE. 2001.) GOTO 851
      IF (SCALE .LE. 5001.) GOTO 852
      IF (SCALE .LE. 25005.) GOTO 853
      IF (SCALE .LE. 100010.) GOTO 854
C
C COPE WITH SCALE GREATER THAN 100000. HERE
      METRES=.FALSE.
      DFACT=10000
      SPCE=100000
      TXSP=10000
      GOTO 855
C
C SCALE .LE. 100000,SCALE .GT. 25000
  854 METRES=.FALSE.
      DFACT=1000
      TXSP=1000
      SPCE=10000
      GOTO 855
C
C SCALE .LE. 25000,SCALE .GT. 5000
  853 DFACT=250
      TXSP=250
      SPCE=1000
      GOTO 855
C
C SCALE .LE. 5000,SCALE .GT. 2000
  852 DFACT=100
      TXSP=100
      SPCE=500
      GOTO 855
C
C SCALE .LE. 2000
  851 DFACT=100
      LABEL=.FALSE.
      TXSP=10
      SPCE=100
  855 CONTINUE
      IFXMIN=((IXMIN+1)/DFACT)*DFACT
      FXMIN=IFXMIN
      IFXMAX=((IXMAX+DFACT-1)/DFACT)*DFACT
      FXMAX=IFXMAX
      IFYMIN=((IYMIN+1)/DFACT)*DFACT
      FYMIN=IFYMIN
      IFYMAX=((IYMAX+DFACT-1)/DFACT)*DFACT
      FYMAX=IFYMAX
```

```
C GET THE RANGES AND ROUND UP AND DOWN TO THE NEAREST PIP
C GET THE MAXIMUM Y VALUE TO WORK OUT IF THE PLOT CAN POSSIBLY
C FIT ON THE PAGE
      YTMP=FYMAX
      CALL XFORM(YTMP,FYMIN,FACT)
      YTMP=YTMP+2.1
C GET TOTAL WIDTH OF NEW PLOT INCLUDING TICKS AND LABELS
      IF (YTMP .LE. PWIDTH) GOTO 870
C WILL FIT ON PAPER
  863 WRITE(5,230)
  230 FORMAT(' PLOT WILL NOT FIT'/
     . ' CHANGE X AND Y RANGES (R) OR SCALE (S)'/' ?')
      READ(5,229) AOK
      IF (AOK .EQ. AR) GOTO 860
C NEW VALUES FOR X AND Y RANGES
      IF (AOK .NE. AS) GOTO 863
      GOTO 862
C EITHER NEED A NEW SCALE OR HAVE AN INVALID OPTION
  870 CONTINUE
C
      NSURV=1
C THE COUNTER FOR THE NUMBER OF DIFFERENT SURVEY NUMBERS
      REWIND 4
C INPUT FILE
      SURV=.FALSE.
      WRITE(5,231)
  231 FORMAT(' NUMBER OF DECIMAL PLACES (0-6)?'/' I')
      READ(5,*) IPLCE
      IF (IPLCE .EQ. 0) IPLCE=-1
C CONVERT 0 DEC. PLACES TO -1 FOR CONVENTION USED IN NUMBER
      FACTMK=1.0
C THE FACTOR FOR SCALING THE SYMBOLS
  232 WRITE(5,233)
  233 FORMAT(' SMALL SYMBOLS? (Y OR N)')
      READ(5,229) AOK
      IF (AOK .EQ. NO) GOTO 234
      IF (AOK .NE. YES) GOTO 232
      FACTMK=0.5
  234 CONTINUE
      WRITE(5,105)
  105 FORMAT(' ENTER MAP TITLE - TERMINATE BY &')
      READ(5,106) TITLE
  106 FORMAT(40A2)
   25 WRITE(5,109)
  109 FORMAT(' OPTIONS'/
     . '  1  LOCATIONS'/
     . '  2  LOCATIONS WITH CIRCLES ROUND NOMINAL POSITION'/
     . '  3  STATION NUMBERS'/
     . '  4  GRAVITY ANOMALY'/
     . '  5  STATION NUMBER AND GRAVITY ANOMALY'/' O')
      READ(5,*) IOPT
      IF (IOPT .LE. 0) GOTO 25
      GOTO (802,801,802,802,802),IOPT
      GOTO 25
C READ THE OPTION AND GO TO THE RIGHT PALCE
```

```
  801 WRITE(5,111)
  111 FORMAT(' XXXXSPACE YYYYSPACE  IN METRES')
      READ(5,*) XSP,YSP
C GET THE NOMINAL STATION SPACING
      WRITE(5,112)
  112 FORMAT(' RRRRADIUS OF CIRCLE IN METRES')
      READ(5,*) ERRRAD
C NOW HAVE ALL ADDITIONAL DETAILS NEEDED FOR OPTION 2
C
  802 WRITE(6,108) TITLE,XMIN,XMAX,YMIN,YMAX,SCALE
  108 FORMAT(1X,40A2/'  BETWEEN',F9.1,' AND',F9.1,' M E AND',
     .  F9.1,' AND',F9.1,' M N'/' SCALE 1:',F8.0)
      IF (YMXPLT+YTMP .LE. PWIDTH .AND. YMXPLT .NE. 0.0) GOTO 23
C FITS ON THE PAPER SO BRANCH
      IF (YMXPLT .GT. 0.) CALL RSTR(1)
C INDEX TO THE NEXT PLOT
      IF (YMXPLT .EQ. 0.) CALL INITAL(9,200,36,1,0,0)
C INITIALLISE PLOTTER
      CALL PLOT(0.5,2.,-3)
C MOVE TO 0.5,2.0 AND CALL IT 0.0,0.0
      YMXPLT=YTMP
      YMXLST=YTMP
      GOTO 24
C
C NOW FOR A PLOT THT FITS ON THE WIDTH AVAILABLE
   23 CALL PLOT(0.0,YMXLST,-3)
C MOVE TO ABOVE THE LAST PLOT AND CALL IT 0.0,0.0
      YMXPLT=YMXPLT+YTMP
      YMXLST=YTMP
C UPDATE YMXPLT
C
   24 CONTINUE
C
C NOW FOR SOME AXES
      DX=IFXMIN-TXSP
   10 DX=DX+TXSP
      XTMP=DX
      CALL XFORM(XTMP,FXMIN,FACT)
      CALL PLOT(XTMP,0.,2)
C MOVE TO START OF PIP
      LINE=MOD(DX,SPCE) .EQ. 0
C CHECK TO SEE IF NEED TO DRAW LINE
      CALL XLABEL(DX,FYMIN,FYMAX,FACT,XTMP,LABEL,LINE,METRES)
      IF (IFXMAX .GT. DX) GOTO 10
C GO ALONG BOTTOM AND FINISH AT A PIP WITH DX=FXMAX,XTMP=FXMAX
  IN PLOTTER INCHES
C
C UP R.H.S.
      DY=IFYMIN-TXSP
   11 DY=DY+TXSP
      YTMP=DY
      CALL XFORM(YTMP,FYMIN,FACT)
      CALL PLOT(XTMP,YTMP,2)
      LINE=MOD(DY,SPCE) .EQ. 0
C CHECK TO SEE IF MULTIPLE
```

```
      CALL YLABEL(DY,FXMIN,FXMAX,FACT,YTMP,LABEL,LINE,METRES)
      IF (IFYMAX .GT. DY) GOTO 11
C NOW GONE UP R.H.S.  DY=FYMAX
C                       YTMP=FYMAX IN PLOTTER COORDS.
C
      CALL PENDN
   12 XTMP=DX
      CALL XFORM(XTMP,FXMIN,FACT)
      CALL PLOT(XTMP,YTMP,1)
      CALL PLOT(XTMP,YTMP+0.1,1)
      CALL PLOT(XTMP,YTMP,1)
      DX=DX-TXSP
      IF (DX .GE. IFXMIN) GOTO 12
C DRAWN PIPS ALONG TOP
C
   14 YTMP=DY
      CALL XFORM(YTMP,FYMIN,FACT)
      CALL PLOT(0.,YTMP,1)
      CALL PLOT(-0.1,YTMP,1)
      CALL PLOT(0.,YTMP,1)
      DY=DY-TXSP
      IF (DY .GE. IFYMIN) GOTO 14
C DRAWN PIPS DOWN L.H.S.
      CALL PENUP
      CALL PWRITE(0.5,-1.0,0.28,0.,TITLE)
C THE TITLE IS PACKED BY THE READ
C WRITE THE TITLE
C
      IF (IOPT .NE. 2) GOTO 31
C DONT NEED TO DRAW CIRCLES ROUND NOMINAL POSITION SO GO ELSEWHERE
      ERRRAD=ERRRAD*FACT
C GET RADIUS IN PLOTTER INCHES
      IXBOT=IXMIN
      IYBOT=IYMIN
C STARTING VALUES FOR NOMINAL POSITIONS
   16 X=IXBOT
      CALL XFORM(X,FXMIN,FACT)
C TRANSFORM TO INCHES
   17 Y=IYBOT
      CALL XFORM(Y,FYMIN,FACT)
C PLOTTER INCHES
      CALL CIRCLE(X,Y,ERRRAD)
C USE THE INBUILT CIRCLE SOFTWARE
      IYBOT=IYBOT+YSP
      IF (IYBOT .LE. IYMAX) GOTO 17
C ALLOW SOME MARGIN FOR THE REPEATED ADDITIONS
      IYBOT=IYMIN
      IXBOT=IXBOT+XSP
      IF (IXBOT .LE. IXMAX) GOTO 16
C CHECK TO SEE STILL IN X RANGE
C DONT TRY TO OPTIMISE PEN MOVEMENT IN THE ABOVE
C
C NOW SET UP THE OTHER OPTIONS
   31 YBOT=0.0
      IF (IOPT .EQ. 5) YBOT=0.07
```

```
C OFFSET FOR LABELLING NUMBER AND GRAVITY
C NOW TO GO THROUGH THE DATA
   30 READ(4,104,END=50) SURVNO,STATNO,X,Y,BA
  104 FORMAT(I5,1X,I4,2F9.1,36X,F8.2)
      NTOT=NTOT+1
C INCREMENT THE TOTAL NUMBER OF STATIONS
      IF (X .LT. XMIN .OR. X .GT. XMAX .OR. Y .LT. YMIN
     . .OR. Y .GT. YMAX) GOTO 30
C CHECK TO SEE IF IN AREA
      NPLOT=NPLOT+1
C INCREMENT THE NUMBER OF STATIONS ACTUALLY PLOTTED
      WRITE(6,107) SURVNO,STATNO,X,Y,BA
  107 FORMAT(I5,'.',I4,2F10.1,F10.2)
C WRITE ALL PLOTTED STATIONS TO PRINTER
      IF (SURV) GOTO 53
      SURV=.TRUE.
      SURVEY(1)=SURVNO
   53 CONTINUE
      DO 15 ISYMB=1,NSURV
      IF (SURVEY(ISYMB) .EQ. SURVNO) GOTO 20
C ALREADY KNOW THIS NUMBER
   15 CONTINUE
      NSURV=NSURV+1
      ISYMB=NSURV
      SURVEY(NSURV)=SURVNO
C ADD NEW SURVEY NUMBER TO INDEX
   20 CONTINUE
      ASURV=STATNO
C GOT STATION NUMBER ALMOST AS AN INTEGER
      CALL XFORM(X,FXMIN,FACT)
      CALL XFORM(Y,FYMIN,FACT)
      CALL PLOT(X,Y,1)
      CALL FACTOR(FACTMK)
      IF (ISYMB .GT. 14) ISYMB=14
C ONLY HAVE 14 SYMBOLS AVAILABLE
      CALL MARKER(ISYMB)
      CALL FACTOR(1.)
      IF (IOPT .LE. 2) GOTO 30
C ONLY WANTED THE MARKER AND THE CIRCLE PERHAPS
      IF (IOPT .EQ. 3 .OR. IOPT .EQ. 5) CALL
     . NUMBER(X+0.07,Y+YBOT,FACTMK*0.07,ASURV,0.,-1)
      IF (IOPT .GE. 4) CALL
     . NUMBER(X+0.07,Y-YBOT,FACTMK*0.07,BA,0.,IPLCE)
      GOTO 30
C PLOTTED AND LABELLED STATION
   50 CONTINUE
C END OF RUN WHEN GET TO HERE SO TIDY UP A LITTLE
      IF (.NOT. SURV) GOTO 54
C NO STATIONS IN AREA
      IF (NSURV .EQ. 1 .AND. SURVEY(1) .EQ. 0) GOTO 54
C HAD DATA WITHOUT SURVEY NUMBERS
      X=0.5
      Y=-1.2
      DO 51 I=1,NSURV
      IMIN=12000
```

```
      DO 52 J=1,NSURV
      IF (SURVEY(J) .EQ. 0) GOTO 52
C ALREADY AND PRINTED IN ORDER
      IF (SURVEY(J) .GT. IMIN) GOTO 52
C ALREADY LOOKING AT AN EARLIER SURVEY
      IMIN=SURVEY(J)
      ISYMB=J
   52 CONTINUE
      CALL PLOT(X,Y,1)
      J=ISYMB
      IF (ISYMB .GT. 14) ISYMB=14
      CALL MARKER(ISYMB)
      ASURV=IMIN
      CALL NUMBER(X+0.15,Y-0.07,.14,ASURV,0.,-1)
      SURVEY(J)=0
      X=X+1.0
C AND SPACE ALONG A BIT
   51 CONTINUE
   54 WRITE(7,227) NTOT,NPLOT
  227 FORMAT(I7,' STATIONS ON FILE'/I7,' STATIONS PLOTTED')
C WRITE THE NUMBER OF STATIONS PLOTTED ON LU 7
      WRITE(5,228)
  228 FORMAT(' MORE PLOTS?')
      READ(5,229) AOK
  229 FORMAT(A1)
      IF (AOK .EQ. YES) GOTO 21
      IF (AOK .NE. NO) GOTO 54
      CALL RSTR(2)
      WRITE(6,235)
  235 FORMAT(1X)
      STOP
      END
      SUBROUTINE XFORM(X,XBASE,FACT)
C FIND THE POSITION OF X RELATIVE TO XBASE IN PLOTTER INCHES
C AND RETURNS THE ANSWER IN X!!!!!!!!!!
      X=(X-XBASE)*FACT
      RETURN
      END
      SUBROUTINE XLABEL(IX,YMIN,YMAX,FACT,XT,LABEL,LINE,METRES)
      LOGICAL LABEL,LINE,METRES
      REAL TITLE(2)
      DATA TITLE/3HM E,4HKM E/
      CALL PLOT(XT,-0.1,1)
      IF (.NOT. LABEL .AND. .NOT. LINE) GOTO 10
C ONLY WANT PIP
      IIX=IX
      IF (.NOT. METRES) IIX=IIX/1000
      CALL NUMPRT(XT-0.035,-0.15,IIX,270.)
      IF (.NOT. LINE) GOTO 10
      CALL WHERE(XX,YY,AX)
C FIND THE PEN
      CALL MOVREL
      IF (METRES) CALL SYMBOL(XX,YY,0.07,TITLE(1),270.,3)
      IF (.NOT. METRES) CALL SYMBOL(XX,YY,0.07,TITLE(2),270.,4)
      CALL MOVABS
```

```
      YY=YMAX
      CALL XFORM(YY,YMIN,FACT)
      CALL PLOT(XT,YY+0.1,1)
      CALL PENDN
   10 CALL PLOT(XT,0.,1)
      RETURN
      END
      SUBROUTINE YLABEL(IY,XMIN,XMAX,FACT,YT,LABEL,LINE,METRES)
      LOGICAL LINE,LABEL,METRES
      REAL TITLE(2)
      DATA TITLE/3HM N,4HKM N/
C USE FOR LABELLING IN SYMBOL
      X=XMAX
      CALL XFORM(X,XMIN,FACT)
      CALL PLOT(X+0.1,YT,1)
      IF (.NOT. LABEL .AND. .NOT. LINE) GOTO 10
C ONLY WANTED PIP
      IIY=IY
      IF (.NOT. METRES) IIY=IIY/1000
      CALL NUMPRT(X+0.15,YT-0.035,IIY,0.0)
      IF (.NOT. LINE) GOTO 10
      CALL WHERE(XX,YY,AY)
C FIND PEN
      CALL MOVREL
      IF (METRES) CALL SYMBOL(XX,YY,0.07,TITLE(1),0.0,3)
      IF (.NOT. METRES) CALL SYMBOL(XX,YY,0.07,TITLE(2),0.,4)
      CALL MOVABS
      CALL PLOT(-0.1,YT,1)
      CALL PENDN
   10 CALL PLOT(X,YT,1)
      RETURN
      END
      SUBROUTINE NUMPRT(X,Y,IVAR,ANG)
C TO WRITE THE INTEGER IVAR WITH NO LEADING BLANKS
C STARTING AT X,Y AT AN ANGLE ANG
C A HEIGHT OF 0.07 IS ASSUMED
      INTEGER*4 BUFF(3)
      ENCODE(BUFF,100) IVAR
  100 FORMAT(I10,'&')
      K=0
      DO 10 I=1,11
      CALL ILBYTE(IB,BUFF,I-1)
      IF (IB .EQ. 32) GOTO 10
C SKIP SPACES
      CALL ISBYTE(IB,BUFF,K)
      K=K+1
   10 CONTINUE
      CALL PWRITE(X,Y,0.07,ANG,BUFF)
      RETURN
      END
```

APPENDIX 2

Sample of output, 1:250 000



BOTHWELL

8151

5 cm

APPENDIX 3

Sample of output, 1:100 000



APPENDIX 4

Sample of output, 1:50 000



BOTHWELL

8151

62-15

5 cm

APPENDIX 5

Sample of output, 1:25 000



APPENDIX 6

Sample of output, 1:5 000


4332
-15.9
295600
295500M N
295400
295300
295200
295100
295000M
294900
4341
-12.4
62-18
500900
501000M E
501100
501200
501300
501400
501500M E
501600
501700
501800
501900
502000M E
502100
502200
BØTHWELL
8151
5 cm

APPENDIX 7

Sample of output, 1:2 500



APPENDIX 8

Sample of output, 1: 2 000